Selectel provider

The Selectel provider is used to interact with the Selectel resources. The provider needs the Selectel API key token to authorize its requests.

Use the navigation to the left to read about the available resources.

Example Usage

```
# Configure the Selectel Provider
provider "selectel" {
  token = "SELECTEL_API_TOKEN_KEY"
}

# Create a project
resource "selectel_vpc_project_v2" "project_1" {
  # ...
}
```

Configuration Reference

The following arguments are supported:

- `token` - (Required) The Selectel API key token. If omitted, the `SEL_TOKEN` environment variable is used.
- `endpoint` - (Optional) The Selectel VPC endpoint. Needed only if this provider is used for tests environment. If omitted, the provider will use the official Selectel VPC endpoint automatically.

Additional Logging

To enable debug logging, set the `TF_LOG` environment variable to `DEBUG`:

```
$ env TF_LOG=DEBUG terraform apply
```

Testing and Development

In order to run the Acceptance Tests for development you need to set the `SEL_TOKEN` environment variable:

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN TF_ACC=1 go test -v ./selectel/...
```

Please create an issue describing a new feature or bug prior creating a pull request.

selectel_vpc_crossregion_subnet_v2

> **WARNING**: this resource has been removed because Selectel VPC Resell V2 API deprecated usage of crossregion subnets.

Manages a V2 Cross-region subnet resource within VPC Selectel VPC.

Example Usage

```
resource "selectel_vpc_project_v2" "project_1" {
  auto_quotas = true
}

resource "selectel_vpc_crossregion_subnet_v2" "crossregion_subnet_1" {
  project_id = "${selectel_vpc_project_v2.project_1.id}"
  cidr = "192.168.200.0/24"
  regions {
    region = "ru-1"
  }
  regions {
    region = "ru-3"
  }
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new Cross-region subnet.
- `regions` - (Required) An array of regions where the Cross-region subnet resides. Changing this creates a new Cross-region subnet. The structure is described below.
- `cidr` - (Required) A cross-region subnet CIDR representation. Changing this creates a new Cross-region subnet.

The `regions` block supports:

- `region` - (Required) A region of where the Cross-region subnet resides. Changing this creates a new Cross-region subnet.

Attributes Reference

The following attributes are exported:

- `servers` - Shows information about servers that use this Cross-region subnet. Contains `id`, `name` and `status` fields.

- `status` - Shows if the Cross-region subnet is used or not.
- `subnets` - Shows information about OpenStack Networking subnets that use this Cross-region subnet. Contains `cidr`, `network_id`, `project_id`, `region`, `subnet_id`, `vlan_id` and `vtep_ip_address` fields.
- `vlan_id` - Shows id of the associated VLAN in the OpenStack Networking service for this Cross-region subnet.

Import

Cross-region subnets can be imported using the `id`, e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_crossregion_subnet_v2.crossregion_subnet
_1 2060
```

selectel_vpc_floatingip_v2

Manages a V2 floating IP resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_floatingip_v2" "floatingip_1" {
  project_id = "887e5e35458d4ee38a6ae0543555dec5"
  region     = "ru-1"
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new floating IP.
- `region` - (Required) A region of where the floating IP resides. Changing this creates a new floating IP.

Attributes Reference

The following attributes are exported:

- `port_id` - Contains id of the Networking service port.
- `floating_ip_address` - Contains floating IP address.
- `fixed_ip_address` - Contains internal IP address of the Networking service port.
- `status` - Shows if the license is used or not.
- `servers` - Shows information about servers that use this floating IP. Contains `id`, `name` and `status` fields.

Import

Floating IPs can be imported using the `id`, e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_floatingip_v2.floatingip_1 aa402146-d83e
-4c8c-8b74-1f415d4b8253
```

selectel_vpc_keypair_v2

Manages a V2 keypair resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_user_v2" "user_1" {
  password = "secret"
}

resource "selectel_vpc_keypair_v2" "keypair_tf_acc_test_1" {
  public_key = "${file("~/.ssh/id_rsa.pub")}"
  user_id    = "${selectel_vpc_user_v2.user_1.id}"
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) Name of the keypair. Changing this creates a new keypair.
- `public_key` - (Required) A pregenerated OpenSSH-formatted public key. Changing this creates a new keypair.
- `regions` - (Optional) List of region names where keypair is need to be created. Keypair will be created in all available regions if omitted. Changing this creates a new keypair.
- `user_id` - (Required) An associated Selectel VPC user. Changing this creates a new keypair.

Attributes Reference

There are no additional attributes for this resource.

Import

Keypairs can be imported by specifying `user_id` and `name` arguments, separated by a forward slash:

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_keypair_v2.keypair_1 <user_id>/<name>
```

selectel_vpc_license_v2

Manages a V2 license resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_license_v2" "license_windows_2016_standard" {
  project_id = "887e5e35458d4ee38a6ae0543555dec5"
  region     = "ru-2"
  type       = "license_windows_2012_standard"
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new license.
- `region` - (Required) A region of where the license resides. Changing this creates a new license.
- `type` - (Required) The type of license. Changing this creates a new license.

Attributes Reference

The following attributes are exported:

- `status` - Shows if the license is used or not.
- `servers` - Shows information about servers that use this license. Contains `id`, `name` and `status` fields.

Import

Licenses can be imported using the `id`, e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_license_v2.license_1 4123
```

selectel_vpc_project_v2

Manages a V2 project resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_project_v2" "kubernetes_cluster" {
  name       = "kubernetes_cluster"
  custom_url = "kubernetes-cluster-123.selvpc.ru"
  theme = {
    color = "2753E9"
  }
  quotas {
    resource_name = "compute_cores"
    resource_quotas {
      region = "ru-3"
      zone = "ru-3a"
      value = 12
    }
  }
  quotas {
    resource_name = "compute_ram"
    resource_quotas {
      region = "ru-3"
      zone = "ru-3a"
      value = 20480
    }
  }
  quotas {
    resource_name = "volume_gigabytes_fast"
    resource_quotas {
      region = "ru-3"
      zone = "ru-3a"
      value = 100
    }
  }
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) The name of the project.
- `custom_url` - (Optional) The custom url for the project. Needs to be the 3rd-level domain for the `selvpc.ru`. Example: `terraform-project-001.selvpc.ru`.
- `theme` - (Optional) An additional theme settings for this project. The structure is described below.
- `auto_quotas` - (Optional) A boolean parameter that specifies if project should get automatically calculated quotas.

- `quotas` - (Optional) An array of desired quotas for this project. The structure is described below.

The `theme` block supports:

- `color` - (Optional) A background color in hex format.
- `logo` - (Optional) An url of the project header logo.

The `quotas` block supports:

- `resource_name` - (Required) A name of the billing resource to set quotas for.
- `resource_quotas` - (Required) An array of desired billing quotas for this particular resource. The structure is described below.

The `resource_quotas` block supports:

- `region` - (Optional) A Selectel VPC region for the resource quota.
- `zone` - (Optional) A Selectel VPC zone for the resource quota.
- `value` - (Required) A value of the resource quota.

Attributes Reference

The following attributes are exported:

- `url` - An url of the Selectel VP project. It is set by the Selectel and can't be changed by the user.
- `enabled` - Shows if project is active or it was disabled by the Selectel.
- `all_quotas` - Contains all quotas. They can differ from the configurable `quota` argument since the project will have all available resource quotas automatically applied.

Import

Projects can be imported using the `id` , e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_project_v2.project_1 0a343062504b4d06a0f
ac375e466db25
```

selectel_vpc_role_v2

Manages a V2 role resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_project_v2" "project_1" {
  auto_quotas = true
}

resource "selectel_vpc_user_v2" "user_1" {
  password    = "secret"
}

resource "selectel_vpc_role_v2" "role_tf_acc_test_1" {
  project_id = "${selectel_vpc_project_v2.project_1.id}"
  user_id    = "${selectel_vpc_user_v2.user_1.id}"
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new role.
- `user_id` - (Required) An associated Selectel VPC user. Changing this creates a new role.

Attributes Reference

There are no additional attributes for this resource.

Import

Roles can be imported by specifying `project_id` and `user_id` arguments, separated by a forward slash:

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_role_v2.role_1 <project_id>/<user_id>
```

selectel_vpc_subnet_v2

Manages a V2 subnet resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_project_v2" "project_1" {
  auto_quotas = true
}

resource "selectel_vpc_subnet_v2" "subnet_1" {
  project_id    = "${selectel_vpc_project_v2.project_1.id}"
  region        = "ru-3"
  ip_version    = "ipv4"
  prefix_length = 29
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new subnet.
- `region` - (Required) A region of where the subnet resides. Changing this creates a new subnet.
- `prefix_length` - (Optional) A prefix length of the subnet. Defaults to 29. Changing this creates a new subnet.
- `ip_version` - (Optional) A version of the IP protocol of the subnet. Defaults to "ipv4". Changing this creates a new subnet.

Attributes Reference

The following attributes are exported:

- `cidr` - Shows subnet CIDR representation.
- `network_id` - Shows associated OpenStack Networking service network ID.
- `subnet_id` - Shows associated OpenStack Networking service subnet ID.
- `status` - Shows if the subnet is used or not.
- `servers` - Shows information about servers that use this subnet. Contains `id`, `name` and `status` fields.

Import

Subnets can be imported using the `id`, e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_subnet_v2.subnet_1 2060
```

selectel_vpc_token_v2

Manages a V2 token resource within Selectel VPC.

ID of this resource can be used within the OpenStack API Identity service as the `X-Auth-Token` value.

Example Usage

```
resource "selectel_vpc_project_v2" "project_1" {
  auto_quotas = true
}

resource "selectel_vpc_token_v2" "token_tf_acc_test_1" {
  project_id = "${selectel_vpc_project_v2.project_1.id}"
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Optional) An associated Selectel VPC project. Changing this creates a new token.
- `account_name` - (Optional) An associated Selectel VPC account. Changing this creates a new token.

Attributes Reference

There are no additional attributes for this resource.

Import

Tokens can't be imported at this time.

selectel_vpc_user_v2

Manages a V2 user resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_user_v2" "user_1" {
  password = "verysecret"
  enabled  = true
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) Name of the user. Changing this updates the name of the existing user.
- `password` - (Required) Password of the user. Changing this updates the password of the existing user.
- `enabled` - (Optional) Enabled state of the user. Changing this updates the enabled state of the existing user.

Attributes Reference

There are no additional attributes for this resource.

Import

Users can't be imported at this time.

selectel_vpc_vrrp_subnet_v2

Manages a V2 VRRP subnet resource within Selectel VPC.

Example Usage

```
resource "selectel_vpc_project_v2" "project_1" {
  auto_quotas = true
}

resource "selectel_vpc_vrrp_subnet_v2" "vrrp_subnet_1" {
  project_id    = "${selectel_vpc_project_v2.project_1.id}"
  master_region = "ru-1"
  slave_region  = "ru-2"
  ip_version    = "ipv4"
  prefix_length = 29
}
```

Argument Reference

The following arguments are supported:

- `project_id` - (Required) An associated Selectel VPC project. Changing this creates a new VRRP subnet.
- `master_region` - (Required) A master region of where the VRRP subnet resides. Changing this creates a new VRRP subnet.
- `slave_region` - (Required) A slave region of where the VRRP subnet resides. Changing this creates a new VRRP subnet.
- `prefix_length` - (Optional) A prefix length of the VRRP subnet. Defaults to 29. Changing this creates a new VRRP subnet.
- `ip_version` - (Optional) A version of the IP protocol of the VRRP subnet. Defaults to "ipv4". Changing this creates a new VRRP subnet.

Attributes Reference

The following attributes are exported:

- `cidr` - Shows VRRP subnet CIDR representation.
- `subnets` - Shows information about OpenStack Networking subnets that use this VRRP subnet. Contains `network_id`, `subnet_id` and `region` fields.
- `status` - Shows if the VRRP subnet is used or not.
- `servers` - Shows information about servers that use this VRRP subnet. Contains `id`, `name` and `status` fields.

Import

VRRP subnets can be imported using the `id` , e.g.

```
$ env SEL_TOKEN=SELECTEL_API_TOKEN terraform import selectel_vpc_vrrp_subnet_v2.vrrp_subnet_1 2060
```